# Compute installation prefix relative to this file
get_filename_component(_prefix "${CMAKE_CURRENT_LIST_DIR}/../../.." ABSOLUTE)

# Report other info
set(RDKit_INCLUDE_DIRS "${_prefix}/@RDKit_HdrDir@")

set(RDKit_HAS_PYTHON_WRAPPERS @RDK_BUILD_PYTHON_WRAPPERS@)
set(RDKit_HAS_COMPRESSED_SUPPLIERS @RDK_BUILD_COMPRESSED_SUPPLIERS@)
set(RDKit_HAS_CAIRO_SUPPORT @RDK_BUILD_CAIRO_SUPPORT@)
set(RDKit_HAS_FREETYPE_SUPPORT @RDK_BUILD_FREETYPE_SUPPORT@)
set(RDKit_HAS_INCHI_SUPPORT @RDK_BUILD_INCHI_SUPPORT@)
set(RDKit_HAS_AVALON_SUPPORT @RDK_BUILD_AVALON_SUPPORT@)
set(RDKit_HAS_THREADSAFE_SSS @RDK_BUILD_THREADSAFE_SSS@)
set(RDKit_HAS_SLN_SUPPORT @RDK_BUILD_SLN_SUPPORT@)
set(RDKit_HAS_DESCRIPTORS3D @RDK_BUILD_DESCRIPTORS3D@)
set(RDKit_HAS_FREESASA_SUPPORT @RDK_BUILD_FREESASA_SUPPORT@)
set(RDKit_HAS_COORDGEN_SUPPORT @RDK_BUILD_COORDGEN_SUPPORT@)
set(RDKit_HAS_MAEPARSER_SUPPORT @RDK_BUILD_MAEPARSER_SUPPORT@)
set(RDKit_HAS_MOLINTERCHANGE_SUPPORT @RDK_BUILD_MOLINTERCHANGE_SUPPORT@)
set(RDKit_HAS_YAEHMOP_SUPPORT @RDK_BUILD_YAEHMOP_SUPPORT@)
set(RDKit_HAS_STRUCTCHECKER_SUPPORT @RDK_BUILD_STRUCTCHECKER_SUPPORT@)
set(RDKit_HAS_URF_SUPPORT @RDK_USE_URF@)

set(RDKit_USE_OPTIMIZED_POPCNT @RDK_OPTIMIZE_POPCNT@)
set(RDKit_USE_STRICT_ROTOR_DEFINITION @RDK_USE_STRICT_ROTOR_DEFINITION@)
set(RDKit_USE_BOOST_VERSION @Boost_VERSION_STRING@)
set(RDKit_USE_BOOST_COMPONENTS headers)
set(RDKit_USE_BOOST_IOSTREAMS @RDK_USE_BOOST_IOSTREAMS@)
if (RDKit_USE_BOOST_IOSTREAMS)
    set(RDKit_USE_BOOST_COMPONENTS ${RDKit_USE_BOOST_COMPONENTS} iostreams)
endif()
set(RDKit_USE_BOOST_SERIALIZATION @RDK_USE_BOOST_SERIALIZATION@)
if (RDKit_USE_BOOST_SERIALIZATION)
    set(RDKit_USE_BOOST_COMPONENTS ${RDKit_USE_BOOST_COMPONENTS} serialization)
endif()

# Find the RDKit dependencies
include(CMakeFindDependencyMacro)

find_dependency(Threads REQUIRED)

find_dependency(Boost @Boost_VERSION_STRING@ COMPONENTS ${RDKit_USE_BOOST_COMPONENTS})

if(RDKit_HAS_DESCRIPTORS3D)
    find_dependency(Eigen3 REQUIRED NO_MODULE)
endif()

# Import the targets
include("${CMAKE_CURRENT_LIST_DIR}/@RDKit_EXPORTED_TARGETS@.cmake")
